

Programare funcțională

Introducere în programarea funcțională folosind Haskell
C07

Claudia Chiriță
Denisa Diaconescu

Departamentul de Informatică, FMI, UB

1

Examen parțial - săptămâna viitoare

- **Test grilă pe foaie** în cadrul cursului
- **3 puncte** din nota finală, **15 întrebări grilă, 40 min**
- Nu este obligatoriu și nu se poate reface
- O să afișăm pe Drive repartitia pe ore
- Materiale ajutatoare: suporturile de curs și de laborator, notițe
- **Materiale ajutatoare doar în format fizic!**
- Fără ceasuri, telefoane etc

2

Functori

Map

În cursurile trecute, am văzut funcția

```
map :: (a -> b) -> [a] -> [b]
```

Problemă. Putem generaliza această funcție la alte tipuri parametrizate?

3

Clasa de tipuri Functor

Definiție

```
class Functor f where  
  fmap :: (a -> b) -> f a -> f b
```

Dată fiind o funcție $g :: a \rightarrow b$ și ca $f :: a$, `fmap` produce `cb :: f b` obținută prin transformarea rezultatelor produse de computația ca folosind funcția g (și doar atât!)

Instanță pentru liste

```
instance Functor [] where  
  fmap = map
```

4

Clasa de tipuri Functor

```
class Functor f where  
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru constructorul de tipuri Maybe
fmap :: (a -> b) -> Maybe a -> Maybe b

```
instance Functor Maybe where  
  fmap f Nothing = Nothing  
  fmap f (Just x) = Just (f x)
```

5

Clasa de tipuri Functor

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru constructorul de tipuri **Either e**
fmap :: (a -> b) -> Either e a -> Either e b

```
instance Functor (Either e) where
  fmap _ (Left x) = Left x
  fmap f (Right y) = Right (f y)
```

6

Clasa de tipuri Functor

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

```
data Arbore a = Nil
              | Nod a (Arbore a) (Arbore a)
```

Instanță pentru constructorul de tipuri **Arbore**
fmap :: (a -> b) -> Arbore a -> Arbore b

```
instance Functor Arbore where
  fmap f Nil = Nil
  fmap f (Nod x l r) = Nod (f x) (fmap f l) (fmap f r)
```

7

Clasa de tipuri Functor

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Tipul funcțiilor de sursă dată **t -> a** (parametric în a)

Instanță pentru tipul funcție
fmap :: (a -> b) -> (t -> a) -> (t -> b)

```
instance Functor (->) a where
  fmap f g = f . g -- sau, mai simplu, fmap = (.)
```

8

Exemple

```
Prelude> fmap (*2) [1..3]
[2,4,6]
Prelude> fmap (*2) (Just 200)
Just 400
Prelude> fmap (*2) Nothing
Nothing
Prelude> fmap (*2) (+100) 4
208
Prelude> fmap (*2) (Right 6)
Right 12
Prelude> fmap (*2) (Left 135)
Left 135
Prelude> (fmap . fmap) (+1) [Just 1, Just 2, Just 3]
[Just 2, Just 3, Just 4]
```

9

Proprietăți ale functorilor

- Argumentul **f** al lui **Functor f** definește o transformare de tipuri
 - f a** este tipul a transformat prin functorul **f**
- fmap** definește transformarea corespunzătoare a funcțiilor
 - fmap :: (a -> b) -> (f a -> f b)**

Contractul lui **fmap**

- fmap f ca** e obținută prin transformarea rezultatelor produse de computația ca folosind funcția **f** (și doar atât!)
- Abstractizat prin două legi:
 - identitate** **fmap id == id**
 - compunere** **fmap (g . h) == fmap g . fmap h**

10

Invalidarea contractului - identitate

```
data WhoCares a = ItDoesnt
                | Matter a
                | WhatThisIsCalled
                deriving (Eq, Show)
```

Instanță a clasei **Functor** care invalidează condiția de conservare a identității:

```
instance Functor WhoCares where
  fmap _ ItDoesnt = WhatThisIsCalled
  fmap _ WhatThisIsCalled = ItDoesnt
  fmap f (Matter a) = Matter (f a)
```

```
Prelude> fmap id ItDoesnt
WhatThisIsCalled
Prelude> id ItDoesnt
ItDoesnt
```

11

Validarea contractului - identitate

```
data WhoCares a = ItDoesnt
  | Matter a
  | WhatThisIsCalled
  deriving (Eq, Show)
```

Instantă a clasei Functor care validează condiția de conservare a identității:

```
instance Functor WhoCares where
  fmap _ ItDoesnt = ItDoesnt
  fmap _ WhatThisIsCalled = WhatThisIsCalled
  fmap f (Matter a) = Matter (f a)
```

```
Prelude> fmap id ItDoesnt
ItDoesnt
Prelude> id ItDoesnt
ItDoesnt
```

12

Invalidarea contractului - compunere

```
data CountingBad a =
  Heisenberg Int a
  deriving (Eq, Show)
```

Instantă a clasei Functor care invalidează condiția de conservare a compunerii:

```
instance Functor CountingBad where
  fmap f (Heisenberg n a) = Heisenberg (n+1) (f a)
```

```
Prelude> oneWhoKnocks = Heisenberg 0 "Uncle"
Prelude> f = (++ " Jesse")
Prelude> g = (++ " lol")
Prelude> fmap (f . g) oneWhoKnocks
Heisenberg 1 "Uncle lol Jesse"
Prelude> fmap f . fmap g $ oneWhoKnocks
Heisenberg 2 "Uncle lol Jesse"
```

13

Validarea contractului - compunere

```
data CountingBad a =
  Heisenberg Int a
  deriving (Eq, Show)
```

Instantă a clasei Functor care validează condiția de conservare a compunerii:

```
instance Functor CountingBad where
  fmap f (Heisenberg n a) = Heisenberg n (f a)
```

```
Prelude> oneWhoKnocks = Heisenberg 0 "Uncle"
Prelude> f = (++ " Jesse")
Prelude> g = (++ " lol")
Prelude> fmap (f . g) oneWhoKnocks
Heisenberg 0 "Uncle lol Jesse"
Prelude> fmap f . fmap g $ oneWhoKnocks
Heisenberg 0 "Uncle lol Jesse"
```

14

Quiz time!



<https://tinyurl.com/PF2023-C08-Quiz1>

15

Functori aplicativi

Problemă

- Folosind **fmap** putem transforma o funcție $h :: a \rightarrow b$ într-o funcție **fmap** $h :: m a \rightarrow m b$
- Dar ce se întâmplă dacă avem o funcție cu mai multe argumente?
De exemplu, cum trecem de la $h :: a \rightarrow b \rightarrow c$ la $h' :: m a \rightarrow m b \rightarrow m c$
- Putem încerca să folosim **fmap**
- Dar, deoarece $h :: a \rightarrow (b \rightarrow c)$, obținem **fmap** $h :: m a \rightarrow m (b \rightarrow c)$
- Putem aplica **fmap** h la o valoare $ca :: m a$ și obținem **fmap** h $ca :: m (b \rightarrow c)$

16

Problemă

Cum transformăm un obiect din $m (b \rightarrow c)$ într-o funcție $m b \rightarrow m c$?

- $\text{ap} :: m (b \rightarrow c) \rightarrow (m b \rightarrow m c)$, sau, ca operator
- $(\langle * \rangle) :: m (b \rightarrow c) \rightarrow m b \rightarrow m c$

17

Merge pentru funcții cu oricâte argumente

Problemă

Dată fiind o funcție

$f :: a1 \rightarrow a2 \rightarrow a3 \rightarrow \dots \rightarrow a_n \rightarrow a$

și computațiile

$ca1 :: m a1, ca2 :: m a2, \dots, can :: m a_n,$

vrem să „aplicăm” funcția f pe rând computațiilor $ca1, \dots, can$ pentru a obține o computație finală $ca :: m a$.

18

Merge pentru funcții cu oricâte argumente

Date fiind

- $f :: a1 \rightarrow a2 \rightarrow a3 \rightarrow \dots \rightarrow a_n \rightarrow a$
- $ca1 :: m a1, ca2 :: m a2, \dots, can :: m a_n,$
- $\text{fmap} :: (a \rightarrow b) \rightarrow m a \rightarrow m b$
- $(\langle * \rangle) :: m (b \rightarrow c) \rightarrow m b \rightarrow m c$ cu „proprietăți bune”

Atunci

```
fmap f :: m a1 -> m (a2 -> a3 -> ... -> a_n -> a)
fmap f ca1 :: m (a2 -> a3 -> ... -> a_n -> a)
fmap f ca1 <*> ca2 :: m (a3 -> ... -> a_n -> a)
...
fmap f ca1 <*> ca2 <*> ca3 ... <*> can :: m a
```

19

Clasa de tipuri Applicative

```
class Functor m => Applicative m where
  pure :: a -> m a
  (<*>) :: m (a -> b) -> m a -> m b
```

- Orice instanță a lui Applicative trebuie să fie instanță a lui **Functor**
- `pure` transformă o valoare într-o computație minimală care are aceea valoare ca rezultat, și nimic mai mult!
- `(<*>)` ia o computație care produce funcții și o computație care produce argumente pentru funcții și obține o computație care produce rezultatele aplicării funcțiilor asupra argumentelor

20

Clasa de tipuri Applicative

```
class Functor m where
  fmap :: (a -> b) -> m a -> m b
```

```
class Functor m => Applicative m where
  pure :: a -> m a
  (<*>) :: m (a -> b) -> m a -> m b
```

Proprietate importantă

- $\text{fmap } f \, x == \text{pure } f \, \langle * \rangle x$
- Se definește operatorul $(\langle \$ \rangle)$ prin $(\langle \$ \rangle) = \text{fmap}$

21

Functori aplicativi

```
($) :: (a -> b) -> a -> b
(<$>) :: (a -> b) -> m a -> m b
(<*>) :: m (a -> b) -> m a -> m b
```

22

Instante – Maybe

```
class Functor m where
  fmap :: (a -> b) -> m a -> m b

class Functor m => Applicative m where
  pure :: a -> m a
  (<*>) :: m (a -> b) -> m a -> m b

instance Applicative Maybe where
  pure = Just
  Nothing <*> _ = Nothing
  Just f <*> x = fmap f x
```

23

Instante – Maybe

```
Prelude> pure "Hey" :: Maybe String
Just "Hey"
```

Cum concatenam **Just** "Hey" cu **Just** "You!"?

```
Prelude> (++) <$> (Just "Hey ") <*> (Just "You!")
Just "Hey You!"
```

- (++) :: **String** -> (**String** -> **String**)
- **Just** "Hey_" :: **Maybe String**
- (<\$>) :: (a -> b) -> m a -> m b (este fmap)
- (++) <\$> (**Just** "Hey_") :: **Maybe (String -> String)**
- **Just** "You!" :: **Maybe String**
- (<*>) :: m (b -> c) -> m b -> m c
- **Just** "Hey_You!" :: **Maybe String**

24

Instante – Maybe

```
mDiv x y = if y == 0 then Nothing
           else Just (x `div` y)

mF x = (+) <$> pure 4 <*> mDiv 10 x

• (+) :: Int -> Int -> Int
• pure 4 :: Maybe Int
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (+) <$> pure 4 :: Maybe (Int -> Int)
• mDiv :: Int -> Int -> Maybe Int
• mDiv 10 x :: Maybe Int
• (<*>) :: m (b -> c) -> m b -> m c

Prelude> mF 2
Just 9
```

25

Instante – Either

```
class Functor m where
  fmap :: (a -> b) -> m a -> m b

class Functor m => Applicative m where
  pure :: a -> m a
  (<*>) :: m (a -> b) -> m a -> m b

instance Applicative (Either a) where
  pure = Right
  Left e <*> _ = Left e
  Right f <*> x = fmap f x
```

26

Instante – Either

```
Prelude> pure "Hey" :: Either a String
Right "Hey"

Prelude> (++) <$> (Right "Hey ") <*> (Right "You!")
Right "Hey You!"

• (++) :: String -> (String -> String)
• Right "Hey_" :: Either a String
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (++) <$> (Right "Hey_") :: Either a (String -> String)
• Right "You!" :: Either a String
• (<*>) :: m (b -> c) -> m b -> m c
• Right "Hey_You!" :: Either a String
```

27

Instante – Either

```
eDiv x y = if y == 0 then Left "Division by 0!"
           else Right (x `div` y)

eF x = (+) <$> pure 4 <*> eDiv 10 x

• (+) :: Int -> Int -> Int
• pure 4 :: Either String Int
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (+) <$> pure 4 :: Either String (Int -> Int)
• eDiv :: Int -> Int -> Either String Int
• eDiv 10 x :: Either String Int
• (<*>) :: m (b -> c) -> m b -> m c

Prelude> eF 2
Right 9
```

28

Instante – Liste

```
class Functor m where
  fmap :: (a -> b) -> m a -> m b

class Functor m => Applicative m where
  pure :: a -> m a
  (<*>) :: m (a -> b) -> m a -> m b

instance Applicative [] where
  pure x = [x]
  fs <*> xs = [f x | f <- fs, x <- xs]
```

29

Instante – Liste

```
Prelude> pure "Hey" :: [String]
["Hey"]

Prelude> (++) <$> ["Hello ", "Goodbye "] <*> ["world"
, "happiness"]
["Hello world","Hello happiness","Goodbye world","
Goodbye happiness"]

• (++) :: String -> (String -> String)
• ["Hello_", "Goodbye_"] :: [String]
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (++) <$> ["Hello_", "Goodbye_"] :: [String -> String]
• ["world", "happiness"] :: [String]
• (<*>) :: m (b -> c) -> m b -> m c
```

30

Instante – Liste

```
Prelude> (+) <$> [1,2] <*> [3,4]
[4,5,5,6]

• (+) :: Int -> Int -> Int
• [1,2] :: [Int]
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (<*>) :: m (b -> c) -> m b -> m c
```

31

Instante – Liste

```
Prelude> [(+),(*)] <*> [1,2] <*> [3,4]
[4,5,5,6,3,4,6,8]

• (+),(*) :: Int -> Int -> Int
• [(+),(*)] :: [Int -> Int -> Int]
• [1,2] :: [Int]
• (<*>) :: m (b -> c) -> m b -> m c
• [(+),(*)] <*> [1,2] :: [Int -> Int]
• [(+),(*)] <*> [1,2] <*> [3,4] :: [Int]
```

32

Instante – Liste

```
Prelude> filter (>50) $ (*) <$> [2,5,10] <*> [8,10,11]
[55,80,100,110]

• (*) :: Int -> Int -> Int
• [2,5,10] :: [Int]
• (<$>) :: (a -> b) -> m a -> m b (este fmap)
• (*) <$> [2,5,10] :: [Int -> Int]
• (<*>) :: m (b -> c) -> m b -> m c
• (*) <$> [2,5,10] <*> [8,10,11] :: [Int]

Prelude> (*) <$> [2,5,10] <*> [8,10,11]
[16,20,22,40,50,55,80,100,110]
```

33

Quiz time!



<https://tinyurl.com/PF2023-C08-Quiz2>

34

Pe săptămâna viitoare!